Solution to Exercise 2.5.1

From Theorem 2.5.9 we see that $\alpha_{C_1,\ldots,C_m} = \alpha_{C_{\sigma(1)},\ldots,C_{\sigma(m)}}$ for any $\sigma \in \mathrm{S}_m$. Hence the following straightforward GAP-program gives the list of all triples $\boldsymbol{C}$ of conjugacy classes (up to ordering) of $G := \mathrm{J}_4$ satisfying $\alpha_{\boldsymbol{C}} = |G|$.

```
gap> ct := CharacterTable("J4");;  triples := [];;
gap> for i in [1..NrConjugacyClasses(ct)] do
>     for j in [i..NrConjugacyClasses(ct)] do
>      for k in [j..NrConjugacyClasses(ct)] do
>       a := ClassStructureCharTable( ct, [i,j,k] );
>       if a = Size(ct) then Add( triples, [i,j,k] ); fi;
>      od;
>     od;
>    od;
gap> List( triples, , x -> List( x, i -> ClassNames(ct)[i] ) );
[ [ "2a", "4a", "11b" ], [ "2b", "2b", "15a" ], [ "2b", "2b", "23a" ],
  [ "2b", "2b", "29a" ], [ "2b", "2b", "30a" ], [ "2b", "2b", "31a" ],
  [ "2b", "2b", "31b" ], [ "2b", "2b", "31c" ], [ "2b", "2b", "33a" ],
  [ "2b", "2b", "33b" ], [ "2b", "2b", "37a" ], [ "2b", "2b", "37b" ],
  [ "2b", "2b", "37c" ], [ "2b", "2b", "40a" ], [ "2b", "2b", "40b" ],
  [ "2b", "2b", "43a" ], [ "2b", "2b", "43b" ], [ "2b", "2b", "43c" ],
  [ "2b", "2b", "44a" ], [ "2b", "2b", "66a" ], [ "2b", "2b", "66b" ] ]
```

Since a group generated by a pair of involutions is a dihedral group (see Remark 2.5.11) it only remains to show that for $\boldsymbol{C} = (\mathtt{2a}, \mathtt{4a}, \mathtt{11b})$ and $(g_1, g_2, g_3) \in \Sigma_{\boldsymbol{C}}$ the group $H = \langle g_1, g_2, g_3 \rangle$ is properly contained in a maximal subgroup of G isomorphic to $2^{11} : \mathrm{M}_{24}$.

We use the fact that the GAP-library of character tables contains the character tables of the maximal subgroups of J_4 together with the fusion maps. So we can find all maximal subgroups M_i of G (up to conjugacy in G) containing elements of 2a, 4a and 11b and

$$X_i := (\ \alpha^{M_i}_{C_1,C_2,C_3} \mid C_i, C_2, C_3 \in \mathrm{cl}(M_i) \ \text{ with } \ C_1 \subseteq \mathtt{2a},\ C_2 \subseteq \mathtt{4a},\ C_3 \subseteq \mathtt{11b}\ ).$$

```
gap> for name in Maxes(ct) do
>     t := CharacterTable( name );
>     fus := Filtered(ComputedClassFusions(t), x -> x.name = "J4")[1].map;
>     if ForAll(triples[1], i -> i in fus) then Print("\n", name," :   ");
>      for i in Positions(fus, triples[1][1]) do
>       for j in  Positions(fus, triples[1][2]) do
>        for k in  Positions(fus, triples[1][3]) do
>         Print(ClassStructureCharTable(t,[i,j,k])/Size(t),",");
>        od;
>       od;
>      od;
>     fi;
>    od;
```

```
mx1j4 :      0,0,0,0,0,1/2,1/2,0,0,
U3(11).2 :      0,0,0,0,
11+^(1+2):(5x2S4) :      0,
```

Instead of $\alpha^{M_i}_{C_1,C_2,C_3}$ we have displayed $\alpha^{M_i}_{C_1,C_2,C_3}/|M_i|$. We see that there are (up to conjugacy) three maximal subgroups of G containing elements of `2a`, `4a` and `11b`, but the only maximal subgroup which may contain g_1, g_2 and g_3 is `mx1j4`. Note that `mx1j4` is the GAP-name for the largest maximal subgroup M_1 of J_4, which is, in fact, isomorphic to $2^{11} : M_{24}$ (see the ATLAS).

G acts by conjugation on $\Sigma_{\boldsymbol{C}}$ and $\mathrm{Stab}_G((g_1, g_2, g_3)) = \mathbf{C}_G(H)$, where $H := \langle g_1, g_2, g_3\rangle$, as above. We have $|\Sigma_{\boldsymbol{C}}| = |G|$ and $\mathbf{Z}(G) = \{1\}$. If $H = G$ then G would act regularly on $\Sigma_{\boldsymbol{C}}$ which is impossible, because there is $(h_1, h_2, h_3) \in \Sigma_{\boldsymbol{C}} \cap M_1 \times M_1 \times M_1$ and hence $\langle h_1, h_2, h_3\rangle \leq M_1$. Thus $H \neq G$ and we may assume that $H \subseteq M_1$.

Suppose that $H = M_1$. Since $\mathbf{C}_G(M_1) = \{1\}$ we conclude that $\mathrm{Stab}_G((g_1, g_2, g_3)) = \{1\}$ and hence $\mathrm{Stab}_{M_1}((g_1, g_2, g_3)) = \{1\}$ and consequently $\alpha^{M_i}_{C_1,C_2,C_3} \geq |M_1|$ for $(C_1, C_2, C_3) := (g_1^{M_1}, g_2^{M_1}, g_3^{M_1})$, which contradicts the above computation showing that $\alpha^{M_i}_{C_1,C_2,C_3} = \frac{1}{2}|M_1|$.